Notes on OptiM → COSY conversions

J.A. Maloney, TRIUMF, June 15, 2015

This note is designed to summarize a conversion from OPTIM to COSY that was done for collaboration on Optical Stochastic Cooling. Since many other beamline codes define elements in manners similar to both COSY and Optim, this may offer insight to any future conversion attempts. This conversion was done using Cosy Infinity 9.1 (COSY), OptimX (OPTIM) May 10, 2015 build.

OptiM is a linear code calculating transfer matrices for elements using TRANSPORT formalization. No fringe field modelling is included. Accordingly, simulations in COSY should be run with FR 0 (no fringe fields) to achieve comparable results.

COSY is a DA-based code that can accurately calculate transfer maps to arbitrary order. It assumes a planar reference orbit and analytic functions to describe the fields and potentials in the midplane.

Drifts:

Specified in Optim:

o or i L[cm]

Specified in COSY

dl L[m]

Conversion: $L_{cosy} = L_{optim}/100$

Magnetic Quadrupoles:

Specified in Optim:

q L[cm] G[kG/cm] Tilt[deg]

Specified in COSY

mq L[m] Poletip Flux [T] Aperture [m]

Optim does not define an aperture so one must be picked for use in COSY based on a real aperture. This value directly effects COSY's fringe field calculation. Aperture in COSY is specified as the vertical half-gap, or the radius from the optical center to the poletip. COSY polarities can also reversed from Optim. Optim assumes particles have a positive charge (q=1). To account for particles with negative charge, the sign of the quadrupole gradient is changed. In COSY, the reference beam is specified so that it can include negative charges (q= -1) and multiple charges (ex. q=3) for the reference particle. It is also possible to adjust the sign of the poletip flux, but this is not a good practice. In the simulation below, for example, the reference particle is an election so focusing quads in Optim have G<0, in COSY poletip flux

>0. Defocusing quads have opposite signs. COSY has a separate element for skew quads and offsets, but this conversion will assume Tilt = 0 in Optim.

Conversion: $L_{cosy} = L_{optim}/100$

$\text{Poletip Flux}_{cosy} = - G_{optim}*10*\text{Aperture}_{cosy}$

Magnetic Sextupoles:

Specified in Optim:

s L[cm] G[kG/cm/cm] Tilt[deg]

Specified in COSY

mh L[m] Poletip Flux [T] Aperture [m]

Notes on conversion for sextupole are the same as for quadrupole.

Conversion: $L_{cosy} = L_{optim}/100$

$\text{Poletip Flux}_{cosy} = - G_{optim}*10*\text{Aperture}_{cosy}*\text{Aperture}_{cosy}$

Magnetic Dipoles:

Specified in Optim (g represents the entrance edge/G represents the exit edge):

g B[kG] Angle[deg] Effective Length [cm] Tilt[deg]

b or d L[cm] B[kG] G[kG/cm] Tilt[deg]

G B[kG] Angle[deg] Effective Length [cm] Tilt[deg]

Specified in COSY

di Radius [m] Bend Angle [deg] Aperture [m] E1[deg] H1[1/m] E2[deg] H2[1/m]

Optim uses and edge element for the edge focusing effect for some edge angle (α) and effective length (a) is defined using K. Brown's formulization:

$$1/F_x = - \tan \alpha/|R|, \quad 1/F_y = \tan \alpha/|R| - a/R^2, \quad a = g K (1 + \sin^2 \alpha)/(\cos^3 \alpha)$$

In this notation, R is the dipole bend radius (which is used in the COSY description of the magnet), g is the vertical half-gap in the dipole and K is a parameter ranging usually from .5 to 1.0. Optim defines the dipole itself by its central field (B) and the path length (L) taken by the reference particle through the dipole. There are also parameters for a field index and tilt which are assumed to be zero. Adding these is possible, but requires using a different element (Procedure MC) in COSY to define the dipole. Bend direction can be specified by the sign of the central field parameter.

Under COSY's definition, the magnet is defined by the bend radius, bend angle and vertical half-gap aperture. The parameters E1 and E2 represent the edge angles and the entrance and exit of the dipole, respectively. H1 and H2 are available for specifying a second order curvature for the entrance and exit edges of the dipole. These are assumed to be 0 since this effect is not specified in Optim. In COSY, the magnet is initially assumed to bend particle with positive charge clockwise, and the command CB is used to switch bend directions.

Translation between the two codes requires specification and conversion of the reference particle and magnetic rigidity. Both codes specify the reference particle in terms of K.E. ($E) and mass ($M). Here is the specification from energy to magnetic rigidity (Bρ) in Optim, and the relationship to the central dipole field ($B), bend radius ($R), path length ($L), and bend angle ($θ):

$E = ($P^2 + $M^2)^{1/2} - $M

$P = (($E + $M)^2 + $M^2)^{1/2}

$\gamma = 1 + ($E/$M)$

$\beta = (1 – (1/\gamma^2))^{1/2}$

Bρ[kG-cm] = $P*10^{11}/$c where $c = 2.9979 * 10^{10} cm/s

$R[cm] = Bρ/$B[kG]

$θ[deg] = $L*360/2π/$R

$L[cm] = $θ*2π*$R/360

In COSY, the reference particle is defined using its mass, charge and K.E. in MeV. There are preset subroutines for electrons and protons where only the KE is specified. For example:

```
RPE  100
```

The internal variable CONS(CHIM) stores the magnetic rigidity for the reference particle and is automatically calculated by COSY.

The dipole half gap or aperture is not explicitly specified in Optim, so it can be assigned in COSY based on the beam size. In COSY, the dipole gap will be used to calculate the fringe field effects, but since there are no such effects using FR 0 model, the choice is arbitrary. It is also possible to estimate the gap from the effective length formula assuming the effective length specified in Optim is not zero and a value

of the K parameter is selected, but without use of COSY fringe field modelling there is no effect of the output from COSY.

The dipole edge is a bit more complicated. $B should be the same for both the dipole and edge in Optim and Tilt=0. If Effective length for the two edge elements ($Efflength1 and $Efflength2) in Optim are not zero, their total length needs to be added to L_{optim}$ before $θ is calculated since COSY with will adjust the total integrated field automatically to match the bend angle and radius specified. Then the edge angle chosen in Optim can be used directly in COSY. An edge angle of zero represents a pure sector dipole where the reference particle enters and exits normal to the magnet edge.

Conversion: R_{cosy}$ = R_{optim}$/100 = Bρ/$B/100

$θ$_{cosy}$[deg] = ($L$_{optim}$ + $Efflength1 + Effelength2)*360/2π/R_{optim}$

Entrance Angle$_{optim}$ = E1$_{cosy}$

Exit Angle$_{optim}$ = E2$_{cosy}$

Wiggler*:**

Specified in Optim using a series of regularly spaced dipoles with reversing polarity.

Specified in COSY

wi B[T] λ[m] L[m] Aperture[m] k l A

COSY's method of calculating matrices, particularly with the inclusion of fringe fields, would make a wiggler created with a series of dipole impractical and likely not particularly accurate. Instead, COSY has a built in element that will be used instead. B is the same field strength used for the dipoles in Optim (T instead of kG), L is the total length of the wiggler. The wiggler field is given by:

$B_m(x, z) – B_0 \cos (2*\pi*z/\lambda + k*z^2)$



Figure 1: Fit run in Mathematica for the OSC wiggler

The final two terms l and A can either both be set to zero, in which case an internal model for the fringe field is used, or if l = 1, A can specify an array of 10 ENGE coefficients to describe the fringe field falloff (g is the vertical half-gap aperture):

$$B(x, z) = B_m(x, z)/(1 + \exp(a_1 + a_2*z/g + \ldots + a_{10}(z/g)^9)$$

***Note that the wiggler in COSY uses the s-dependent integrator. COSY's functionality to output parameter dependent maps using DA variables will not work with this element.

Beta Function and Twiss Parameters:

Specified in Optim:

Emittance: ex[cm]= ey[cm]= DP/P=

Initial: BetaX[cm]= BetaY[cm]=

AlfaX= AlfaY=

Qx=0 Qy=0

DispersX[cm]= DispersY[cm]=0

DspPrimeX= DspPrimeY=0

X[cm]= Y[cm]= Z[cm]= S[cm]=

tetaX[deg]= tetaY[deg]=0

Specified in COSY

SB dx da r12 dy db r34 dt de r56 dg dz;

In Optim, the initial Twiss parameters for the beam are specified as well as the beam emittance. This is does in a block format as described above. The initial line sets the 2D emittance in the horizontal and vertical planes, as well as the desired momentum spread. The next two lines set α and β twiss parameters in the horizontal and vertical planes. The remaining lines allow for other modifications that are not discussed in this note, and are assumed to be set to zero.

In COSY, the format is to specify a reference particle and various half-widths for the other beam coordinates (dx da r12 dy db r34 dt de r56 dg dz). r12, r34 and r56 are the TRANSPORT notation parameters. The transverse half-widths, dz and da, are in meters. The half-angles, da and db, are in radians. dt is specified as $v_0\gamma/(1 + \gamma)$, de = ΔE/E, dg = Δmass/mass and dz = Δz/z. This conversion will assume dt, r56, dg and dz are zero.

Conversion: $ex_{cosy} = ex_{optim}/100$ $ey_{cosy} = ey_{optim}/100$

$betax_{cosy} = betax_{optim}/100$ $Betay_{cosy} = betay_{optim}/100$

$alfax_{cosy} = alfax_{optim}$ $alfay_{cosy} = alfay_{optim}$

$gammax_{cosy} = (1 + alfax_{cosy}^2)/betax_{cosy}$

$gammay_{cosy} = (1 + alfay_{cosy}^2)/betay_{cosy}$

$dx_{cosy} = Sqrt(ex_{cosy}*betax_{cosy})$ $da_{cosy} = Sqrt(ex_{cosy}*gammax_{cosy})$

$dy_{cosy} = Sqrt(ey_{cosy}*betay_{cosy})$ $db_{cosy} = Sqrt(ey_{cosy}*gammay_{cosy})$

$r12 = - alfax_{cosy}/Sqrt(betax_{cosy}*gammax_{cosy})$

$r34 = - alfay_{cosy}/Sqrt(betay_{cosy}*gammay_{cosy})$

In COSY, the twiss parameter functions are calculated and stored throughout the beamline in the sigma-matrix format. These can be called at any time using the sigma(i,j). The value of the twiss parameter can then be determined by dividing the matrix term by the 2D emittance in that phase space. For example, to find the beta function in the horizontal plane (Bx), you can define the emittance (in the same format as the SB routine), and use code sequencing similar to this:

```
Emitx:=dx*da*COS(ASIN(R12));

Bx:=SIGMA(1,1)/Emitx;
```

This value can be called at any point in the beamline, and the updated value will be calculated.

```
Results from conversion of OSC Chicane from Optim to COSY:
From Element 1(oow) to Element 170(oow) Matrix.
Energy increase[MeV]=0 (From 100 to 100) Delta TetaY=0 (From 0 to 0)
X[cm] Px/P Y[cm] Py/P dL[cm] dP/P

-2.172091e+000 -4.415181e+002  0.000000e+000  0.000000e+000 0.000000e+000 3.956090e-001
-8.420902e-003 -2.172091e+000  0.000000e+000  0.000000e+000 0.000000e+000 1.050217e-003
 0.000000e+000  0.000000e+000 -4.858632e-001 -4.533561e+001 0.000000e+000 0.000000e+000
 0.000000e+000  0.000000e+000  1.685070e-002 -4.858632e-001 0.000000e+000 0.000000e+000
-1.050217e-003 -3.956090e-001  0.000000e+000  0.000000e+000 1.000000e+000 6.010390e-001
 0.000000e+000  0.000000e+000  0.000000e+000  0.000000e+000 0.000000e+000 1.000000e+000

and here is the final matrix from COSY:

  -2.172091     -0.8420897     0.0000000E+00  0.0000000E+00 -0.1044903E-02 100000
  -4.415180     -2.172091      0.0000000E+00  0.0000000E+00 -0.3936075E-02 010000
  0.0000000E+00 0.0000000E+00 -0.4858631      1.685071       0.0000000E+00 001000
  0.0000000E+00 0.0000000E+00 -0.4533560     -0.4858631      0.0000000E+00 000100
  0.0000000E+00 0.0000000E+00  0.0000000E+00  0.0000000E+00  1.000000      000010
  0.3936075E-02 0.1044903E-02  0.0000000E+00  0.0000000E+00  0.5949737E-02 000001

COSY's map represents the transpose of the Optim matrix with the final row omitted since all
terms are 0 or 1 in systems where energy of the reference particle is unchanged. The coefficients
for the columns show the variations in final coordinates (x, a, y, b, and l) based on initial
coordinates (x, a, y, b, l, δ).  This information can be obtained for COSY separately if needed
(for example, if the system contains RF cavities or absorbers).  The initial parameter in the
final row of COSY map represents (dK/K) =(dP/P)(γ + 1)/γ. This adjustment must be considered
before comparing the COSY results with those in Optim.  For the first two terms, the coefficient
```

in COSY should be multiplied by $(\gamma + 1)/\gamma$ to get the corresponding value the Optim matrix. For the (6|5) term in Optim, the COSY result must be multiplied by the square of $(\gamma + 1)/\gamma$.

Example COSY conversion:

```
INCLUDE 'C:\Users\maloneyja\Documents\COSY files\cosy';
PROCEDURE RUN;

VARIABLE MASS 1;  {Centered beam Mass}
VARIABLE DX0 1; VARIABLE DY0 1;  {Beam spot size parameters}
VARIABLE DA0 1; VARIABLE DB0 1;  {Beam angle parameters}
VARIABLE DM 1; VARIABLE DE 1; {Mass and Energy Resolution}
VARIABLE R12 1; Variable R34 1;

Variable BWIG 1; Variable PhiCh 1; variable obj 1;
variable fitvar 1; variable current_length 1; variable betas_on 1;
variable betax 1; variable betay 1; variable alphax 1; variable alphay 1;
variable gammax 1; variable gammay 1; variable dispx 1; variable dispxp 1;
variable Bx 1; variable By 1; variable Ax 1;
variable Ay 1; variable Gx 1; variable Gy 1;
Variable emitx 1; Variable emity 1;
VARIABLE ENVELX 1; VARIABLE ENVELY 1;
variable ymax 1; variable bmax 1; variable xmax 1; variable amax 1;
variable mux 1; variable muy 1;

{************TWISS PARAMETER PROCEDURES*************}

Procedure betas;
Emitx:=DX0*DA0*COS(ASIN(R12));
Emity:=DY0*DB0*COS(ASIN(R34));
Bx:=SIGMA(1,1)/Emitx;
Ax:=-SIGMA(1,2)/Emitx;
Gx:=SIGMA(2,2)/Emitx;
By:=SIGMA(3,3)/Emity;
Ay:=-SIGMA(3,4)/Emity;
Gy:=SIGMA(4,4)/Emity;
ENVELX:= ABS(SQRT((ME(1,1)*DX0)^2 + (ME(1,2)*DA0)^2));
ENVELY:= ABS(SQRT((ME(3,3)*DY0)^2 + (ME(3,4)*DB0)^2));
```

```
If Spos=0;
write 36 'Beam pathlength - Twiss parameters Bx Ax Gx Envelx:';
write 37 'Beam pathlength - Twiss parameters By Ay Gy Envely:';
endif;
write 36 SPOS&Bx&Ax&Gx&Envelx;
write 37 SPOS&By&Ay&Gy&Envely;
endprocedure;

{*************** Matching subroutines ******************}
Procedure verticalmatch;
muy:=ACOS((ME(3,3)+ME(4,4))/2);  Write 6 'muy ' muy;
Betay:=ABS(ME(3,4)/SIN(muy)); Write 6 'betay ' betay;
Gammay:=ABS(ME(4,3)/SIN(muy)); Write 6 'gammay ' gammay;
ymax:=SQRT(Betay*emity);
bmax:=SQRT(Gammay*emity);
R34:=SQRT(1-SQR(Emity/ymax/bmax));
write 6 'match y b r34: ' ymax&bmax&r34;
Dy0:=ymax;
Db0:=bmax;
R34:=SQRT(1-SQR(Emity/ymax/bmax));
Endprocedure;

Procedure Horizontalmatch;  {Matches the initial/final horizontal twiss parameters}
Emitx:=DX0*DA0*COS(ASIN(R12));
mux:=ACOS((ME(1,1)+ME(2,3))/2);  Write 6 'muy ' muy;
Betax:=ABS(ME(1,2)/SIN(mux)); Write 6 'betay ' betay;
Gammax:=ABS(ME(2,1)/SIN(mux)); Write 6 'gammay ' gammay;
xmax:=SQRT(Betax*emitx);
amax:=SQRT(Gammax*emitx);
R12:=SQRT(1-SQR(Emitx/xmax/amax));
write 6 'match v a r12: ' xmax&amax&r12;
Dx0:=xmax;
Da0:=amax;
R12:=SQRT(1-SQR(Emitx/xmax/amax));
```

```
Endprocedure;

{***************  Initial parameters ******************}

PROCEDURE INIT_VAR;  {Sets Initial Parameter Values}
DX0:=.0039756426; DA0:=.001193540686; R12:=-.950803754;
DY0:=.000818011; DB0:=.002075708; R34:=-.500479589;
DM:= 0; DE:=.0006533046353;
endprocedure;

{************ CHICANE LATTICE *****************}

Procedure OSC; fr 0; betas;
write 6 sigma(1,1) sigma(1,2) sigma(2,1) sigma(2,2);
dl .4225; pm 6; betas;
wi BWIG .08 .64 .025 15.5 0 0; pm 6; betas;
dl .01252; pm 6; betas;
dl .10998; pm 6; betas;
dl .2065; pm 6; betas;
mq .15 .143 .025; pm 6; betas;
dl .15; pm 6; betas;
mq .15 -.1734942 .025; pm 6; betas;
dl .15; pm 6; betas;
di .8090354329 7.081986368 .025 0 0 7.081986368 0; pm 6; betas;
cb;
dl .05; pm 6; betas;
mh .03 -.018 0; pm 6; betas;
dl .05; pm 6; betas;
di .8090354329 7.081986368 .025 7.081986368 0 0 0; pm 6; betas;
dl .07; pm 6; betas;
mq .03 0.03 .025; pm 6; betas;
mq .03 0.03 .025; pm 6; betas;
dl .07; pm 6; betas;
di .8090354329 7.081986368 .025 0 0 7.081986368 0; pm 6; betas;
```

```
dl .05; pm 6; betas;
mh .03 -.018 0; pm 6; betas;
dl .05; pm 6; betas;
cb;
di .8090354329 7.081986368 .025 7.081986368 0 0 0; pm 6; betas;
dl .15; pm 6; betas;
mq .15 -.1734942 .025; pm 6; betas;
dl .15; pm 6; betas;
mq .15 .143 .025; pm 6; betas;
dl .2065; pm 6; betas;
dl .10998; pm 6; betas;
dl .01252; pm 6; betas;
wi BWIG .08 .64 .025 15.5 0 0; pm 6; betas;
dl .4225; betas;
Endprocedure;
```

```
ov 1 3 0;
LAX:=0; LCE:=0;
INIT_VAR;
RPE 100;
{SBE 1e-2 1e-2 1e-2;}
SB dx0 da0 r12 dy0 db0 r34 0 de 0 0 0;
um; cr; ER 3 7 1 2 1 3 2 2 ;
ENVEL; PTY 0; bp;
{fit fitvar;}
OSC;
{obj:=sqr(ME(1,2))+sqr(ME(2,1))+sqr(ME(3,4))+sqr(ME(4,3));}
{endfit 1e-10 100 1 obj;
write 6 'fitvar ' fitvar;
write 6 'obj ' obj;}
pm 6; EP; PG -10 -10;
ENDPROCEDURE; RUN; END;
```

Example Optim file:

```
# Corrections
#    Added OSC section
#      oq -> oqd downstream of q15
#     quads q11,12,15 &16 are decoupled from corresponding quads in visinity of OSC section
$E=100;  =>         100
   $P=sqrt(($E+$Me)^2-$Me^2);  =>  100.509704
   $gamma=1+$E/$Me;  =>  196.693414
   $beta=sqrt(1-1/$gamma^2);  => 0.999987076
   $Bro=1.e11/$c*$P;  =>  335.264286
  $Rdip=70;  =>          70
  $Bdip=$Bro/$Rdip;  =>   4.7894898
  $Ldip60=60/180*$PI*$Rdip;  =>  73.3038286
  $Ldip30=30/180*$PI*$Rdip;  =>  36.6519143
#---------------
# OSC
$BdipCh=4.144;  =>       4.144
$LdipCh=10;  =>          10
  $PhiCh=180/$PI*$BdipCh*$LdipCh/$Bro;  =>  7.08198637
# Wiggler parameters
$Bwgl=0;  =>           0
$Lwgl=8 / 4;  =>           2
  $PhiWgl=180/$PI*$Bwgl*$Lwgl/$Bro/2;  =>           0
#chicane center:
$BetaX=4;  =>           4
$BetaY=157;  =>         157
$DispX=66      ;  =>          66
#-----------------
OptiM
Energy[MeV]=100        Mass[MeV]=0.5110034
Emittance: ex[cm]=0.000147  ey[cm]=0.000147  DP/P=0.00065
Initial:       BetaX[cm]=1075.22     BetaY[cm]=45.5199
       AlfaX=3.06915  AlfaY=0.578089  Qx=0  Qy=0
       DispersX[cm]=-217.136  DispersY[cm]=0
       Dsp_PrimeX=1.56091     DspPrimeY=0
       X[cm]=2.837            Y[cm]=0.000           Z[cm]=242.000         S[cm]=0.000
       tetaX[deg]=-180        tetaY[deg]=0
begin lattice. Number of periods=1
oow bWGLhm gWGLm oW1 gWGLp bWGLp gWGLp
oW gWGLm bWGLm gWGLm oW gWGLp bWGLp
gWGLp oW gWGLm bWGLm gWGLm oW gWGLp
bWGLp gWGLp oW gWGLm bWGLm gWGLm oW
gWGLp bWGLp gWGLp oW gWGLm bWGLm gWGLm
oW gWGLp bWGLp gWGLp oW gWGLm bWGLm
gWGLm oW gWGLp bWGLp gWGLp oW gWGLm
bWGLm gWGLm oW gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW gWGLp bWGLp gWGLp
oW gWGLm bWGLm gWGLm oW1 gWGLp bWGLhp
oWend oL qChF1 oqb qChD oqb bChp
gChp obs sOSC obs gChm bChm oqo
qChF
#
qChF oqo bChm gChm obs sOSC obs gChp bChp
oqb  qChD oqb qChF1 oL oWend
#
bWGLhp gWGLp oW1
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW
gWGLm bWGLm gWGLm oW  gWGLp bWGLp gWGLp oW1
gWGLm bWGLhm
#
oow
end lattice
#=====================================
```

```
begin list
oqo     L[cm]=7
obs     L[cm]=5
sOSC    L[cm]=3               S[kG/cm/cm)]=-2.88   Tilt[deg]=0
oqb     L[cm]=15
oL      L[cm]=20.65
oow     L[cm]=42.25
bChp    L[cm]=10              B[kG]=4.144          G[kG/cm]=0            Tilt[deg]=0
gChp    B[kG]=4.144      Angle[deg]=7.08199  EffLen[cm]=0  Tilt[deg]=0
bChm    L[cm]=10              B[kG]=-4.144         G[kG/cm]=0            Tilt[deg]=0
gChm    B[kG]=-4.144     Angle[deg]=7.08199  EffLen[cm]=0  Tilt[deg]=0
# wigler
bWGLp   L[cm]=2               B[kG]=0              G[kG/cm]=0            Tilt[deg]=0
gWGLp   B[kG]=0          Angle[deg]=0  EffLen[cm]=0  Tilt[deg]=0
bWGLm   L[cm]=2               B[kG]=-0             G[kG/cm]=0            Tilt[deg]=0
gWGLm   B[kG]=-0         Angle[deg]=0  EffLen[cm]=0  Tilt[deg]=0
bWGLhp  L[cm]=1               B[kG]=0              G[kG/cm]=0            Tilt[deg]=0
bWGLhm  L[cm]=1               B[kG]=-0             G[kG/cm]=0            Tilt[deg]=0
#
oW      L[cm]=2
oW1     L[cm]=0.626
oWend   L[cm]=10.998

ibpm    L[cm]=9               Tilt[deg]=0
od22    L[cm]=23
oq      L[cm]=6.25
ob      L[cm]=31
o1      L[cm]=90
oqd     L[cm]=12.5

o1a     L[cm]=75
oq2     L[cm]=10

oq0     L[cm]=29.5
oq1     L[cm]=0.5
oq3     L[cm]=20

oq1a    L[cm]=1
oq2a    L[cm]=61
oq3a    L[cm]=1
oq4a    L[cm]=85
#
b30     L[cm]=36.651914       B[kG]=4.7894898      G[kG/cm]=0            Tilt[deg]=0
gb30    B[kG]=4.7894898   Angle[deg]=0  EffLen[cm]=1  Tilt[deg]=0
Gb30    B[kG]=4.7894898   Angle[deg]=0  EffLen[cm]=1  Tilt[deg]=0
#
b60     L[cm]=73.303829       B[kG]=4.7894898      G[kG/cm]=0            Tilt[deg]=0
gb60    B[kG]=4.7894898   Angle[deg]=0  EffLen[cm]=1  Tilt[deg]=0
Gb60    B[kG]=4.7894898   Angle[deg]=0  EffLen[cm]=1  Tilt[deg]=0
#
qChF    L[cm]=3               G[kG/cm]=-0.12        Tilt[deg]=0
qChD    L[cm]=15              G[kG/cm]=0.6939768    Tilt[deg]=0
qChF1   L[cm]=15              G[kG/cm]=-0.572       Tilt[deg]=0
q11c    L[cm]=21              G[kG/cm]=0            Tilt[deg]=0
q10     L[cm]=21              G[kG/cm]=-0.23        Tilt[deg]=0
q9      L[cm]=21              G[kG/cm]=0.23         Tilt[deg]=0
q8      L[cm]=21              G[kG/cm]=0.455        Tilt[deg]=0
q7      L[cm]=21              G[kG/cm]=-0.29        Tilt[deg]=0
q6      L[cm]=21              G[kG/cm]=0.32         Tilt[deg]=0
q5      L[cm]=21              G[kG/cm]=-0.43        Tilt[deg]=0
q4      L[cm]=21              G[kG/cm]=0.405        Tilt[deg]=0
q3      L[cm]=21              G[kG/cm]=0.4404501    Tilt[deg]=0
q2      L[cm]=21              G[kG/cm]=-0.5293225   Tilt[deg]=0
q1      L[cm]=21              G[kG/cm]=0.2129649    Tilt[deg]=0
q20     L[cm]=21              G[kG/cm]=0.2512792    Tilt[deg]=0
q19     L[cm]=21              G[kG/cm]=-0.4229292   Tilt[deg]=0
q18     L[cm]=21              G[kG/cm]=0.2512792    Tilt[deg]=0
q17     L[cm]=21              G[kG/cm]=-0.1366687   Tilt[deg]=0
q16     L[cm]=21              G[kG/cm]=0.213753     Tilt[deg]=0
q15     L[cm]=21              G[kG/cm]=-0.2219317   Tilt[deg]=0
q14     L[cm]=21              G[kG/cm]=0.1604115    Tilt[deg]=0
```

```
end list of elements
#=====================================
4D_BetaBlock
        Beta_X_1[cm]=1075.23   Beta_X_2[cm]=0.
        Alfa_X_1=3.06918       Alfa_X_2=0.
        Beta_Y_1[cm]=0         Beta_Y_2[cm]=45.5198
        Alfa_Y_1=0             Alfa_Y_2=0.578088
        Nu_1[deg]=0.           Nu_2[deg]=0    U=0
        Emit1[cm]=1.e-5.       Emit2[cm]=1.e-5
        DispX[cm]=-217.136            DispY[cm]=0
        DispXpr[cm]=1.56091    DispYpr[cm]=0
End4DBetaBlock
```